

# FIBPlus Developer's Guide

## Part I

### Database connection

To connect to a database (DB) you should use the TpFIBDatabase component. For more details about its properties and methods read FIBPlus help file.

#### *Connection parameters*

Connection parameters are typical for InterBase/Firebird server:

- path to a database file;
- user name and password;
- user role;
- charset;
- dialect;
- client library (gds32.dll for InterBase and fbclient.dll for Firebird).

To set all the properties at once you can use a built-in connection setting dialog (see picture 1).

The dialog «Database Editor» can be invoked from the component context menu (right click on the component) at design-time.

Here you may set all necessary parameters, including getting them from/saving them to Alias. You may also check whether the parameters are correct by using a test connection.




Figure 1. Connection Properties TpFIBDatabase

Similar to the actions in the dialog, you can do the same in the application code.

To connect to a database you should call the Open method or set the Connected property to True. It's also possible to use this code to connect to a database:

```
function Login(DataBase: TpFIBDatabase; dbpath, uname, upass, urole: string):  
Boolean;  
begin  
  if DataBase.Connected then DataBase.Connected := False;  
  with FDataBase.ConnectParams do begin  
    UserName := uname;  
    Password := upass;  
    RoleName := urole;  
  end;  
  DataBase.DBName := dbpath;  
  try DataBase.Connected := True;  
  except  
    on e: Exception do  
      ShowMessage(e.Message);  
  end;  
  Result := DataBase.Connected;  
end;
```

To close the connection either call the Close method or set the Connected property to False. You can also close all datasets and connected transactions at once:

```
procedure Logout(DataBase: TpFIBDatabase);  
var i: Integer;  
begin  
  if not DataBase.Connected then  
    Exit;  
  for i := 0 to DataBase.TransactionCount - 1 do  
    if TpFIBTransaction(DataBase.Transactions[i]).InTransaction then  
      TpFIBTransaction(DataBase.Transactions[i]).Rollback  
  DataBase.CloseDataSets;  
  DataBase.Close;  
end;
```

## *How to create and drop database*

It's very easy to create a new DB. You need to set DB parameters and call the CreateDatabase method:

### **Delphi**

```
with Database1 do begin
  DBParams.Clear;
  DBParams.Add('USER 'SYSDBA' PASSWORD 'masterkey');
  DBParams.Add('PAGE_SIZE = 2048');
  DBParams.Add('DEFAULT CHARACTER SET WIN1251');
  DBName := 'SERV_DB:C:\DB\TEST.IB';
  SQLDialect := 3;
end;

try
  Database1.CreateDataBase;
except
  // Error handling
end;
```

### **C++**

```
Database1->DBParams->Clear();
Database1->DBParams->Add("USER 'SYSDBA' PASSWORD 'masterkey'");
Database1->DBParams->Add("PAGE_SIZE = 2048");
Database1->DBParams->Add("DEFAULT CHARACTER SET WIN1251");
Database1->DBName = "SERV_DB:C:\\DB\\TEST.GDB";
Database1->SQLDialect = 3;
```

### **try**

```
{ Database1->CreateDatabase(); }
catch (...)
{ // Error
}
```

To drop a database, use the DropDatabase method. Note: you should be connected to the database when using this method.

## *Metadata caching*

FIBPlus enables developers to get system information about field tables automatically, set in TpFIBDataSet such field properties as Required (for NOT NULL fields), ReadOnly (for calculated fields) and DefaultExpression (for fields with default database values). This feature is very useful for both programmers and users, because programmers do not need to set property values manually when writing client applications, and users get clearer messages when working with the programme e.g. if any field is NOT NULL, and the user attempts to leave it empty, he will see a message «Field '...' must have a value.». This is more understandable than a system InterBase/Firebird PRIMARY KEY violation error. The same with calculated fields, it is not possible to edit such fields, so FIBPlus will set the ReadOnly property to True for all calculated fields, and the user will not get a vague message on trying to change the field values in TDBGrid.

This feature has one disadvantage, which is revealed on low speed connections. To get information on fields, FIBPlus components execute additional “internal queries” on InterBase/Firebird system tables. If there are many tables in the application, or many fields in these tables, the application work can slow down and net traffic increase. This becomes especially obvious on first query opening, as every open query is followed by the internal queries. On subsequent opening of the query, FIBPlus uses the information, which has already been obtained, but users may notice a slight work slowdown when the application starts.

This is where Metadata caching comes in. TpFIBDatabase enables developers to save metadata information at the client machine and use it during the applications execution and

subsequent executions. The *TCacheSchemaOptions*.property is responsible for this process:

```
TCacheSchemaOptions = class(TPersistent)
  property LocalCacheFile: string;
  property AutoSaveToFile: Boolean .. default False;
  property AutoLoadFromFile: Boolean .. default False;
  property ValidateAfterLoad: Boolean .. default True;
end;
```

The *LocalCacheFile* property sets the name of the local cache file where this information will be saved. *AutoSaveToFile* helps to save cache to the file automatically on closing the application. *AutoLoadFromFile* loads cache from the file. And *ValidateAfterLoad* defines whether it's necessary to check the saved cache after its loading. Besides there is an *OnAcceptCacheSchema* event, where you may define objects, for which you don't need to load the saved information.

This property is also very easy-to-use.

```
with pFIBDatabase1.CacheSchemaOptions do begin
  LocalCacheFile := 'fibplus.cache';
  AutoSaveToFile := True;
  AutoLoadFromFile := True;
  ValidateAfterLoad:= True;
end;
```

To summarize, FIBPlus gathers information about fields accessed in a *TpFIBDataset*. This information is used to set properties on *TField*'s. To reduce the overhead this introduces, the *TpFIBDatabase* can be set to save and load this cache between program executions.

### ***BLOB field caching***

BLOB field caching at the client is one more unique FIBPlus feature. Blobs are unique in InterBase/Firebird compared to other datatypes. When returned in a query, what is actually returned is a BLOB ID. When the field value is required, FIBPlus automatically asks the server for the data that relates to the BLOB ID, i.e. generating at least one more round trip to the server. This can create a performance bottleneck when the same BLOB is retrieved a number of times. BLOB field Caching helps reduce this performance hit.

If *BlobSwapSupport.Active := True*, FIBPlus will automatically save fetched BLOB fields in the defined directory (the *SwapDir* property). By default the *SwapDir* property is equal to *{APP\_PATH}*, that is, it sets the directory with the executed application. You can also set the directory where to save BLOB fields. I.e., *SwapDir := '{APP\_PATH}' + '\BLOB\_FILES'*

There are four events enabling work with this property in *TpFIBDatabase*:

```
property BeforeSaveBlobToSwap: TBeforeSaveBlobToSwap;
property AfterSaveBlobToSwap: TAfterSaveLoadBlobSwap;
property AfterLoadBlobFromSwap: TAfterSaveLoadBlobSwap;
property BeforeLoadBlobFromSwap: TBeforeLoadBlobFromSwap;
```

where

```
TBeforeSaveBlobToSwap = procedure(const TableName, FieldName: string;
RecordKeyValues: array of variant; Stream: TStream; var FileName: string; var
CanSave: boolean) of object;
TAfterSaveLoadBlobSwap = procedure(const TableName, FieldName: string;
RecordKeyValues: array of variant; const FileName: string) of object;
TBeforeLoadBlobFromSwap = procedure(const TableName, FieldName: string;
RecordKeyValues: array of variant; var FileName: string; var CanLoad:
boolean) of object;
```

You do not need to worry about these event handlers unless you want more control over how the BLOB field cache works. Event handlers help to manage BLOB field saving and reading from the disc. In particular you can forbid saving some BLOB field in event handlers depending on the field name and the table name, on other field values, free disc space, etc.

You can also save BLOB fields by using the MinBlobSizeToSwap property. There you can set a minimal BLOB field size for saving at disk..

This technology has a number of limitations:

1. The table must have a primary key.
2. BLOB fields must be read by the TpFIBDataSet component.
3. Your application must monitor the free disk space. For these purposes you can use the even handler BeforeSaveBlobToSwap.
4. Unfortunately all BLOB\_ID's are changed after database backup/restore, so local cache becomes useless and is automatically cleared. After each application connection to the database, FIBPlus automatically opens a special thread, which checks in a separate connection the whole disk cache on BLOB fields. If there are no BLOB fields, the corresponding files are immediately deleted.

### ***Client BLOB filters***

These are user functions which help to handle (encrypt, pack, unpack, etc) BLOB fields at the client transparently for the user application. This feature helps you to pack or code blob fields in a database without changing the client application. FIBPlus has a mechanism of client BLOB filters similar to the one built in InterBase/Firebird. An advantage of a local blob-filter is an ability to decrease network traffic of the application considerably if you pack blob-fields before sending them to and then unpack them after getting to the client. This is done by means of registering two procedures for reading and writing blob-fields in TpFIBDatabase. As a result FIBPlus will automatically use these procedures to handle all blob-fields of the set type in all TpFIBDataSets using one TpFIBDatabase instance.

To illustrate this technology we will write an example of handlers which will pack and unpack each BLOB field and register these handlers. Note: Remember that user BLOB field subtype must be negative, the positive values allocated to InterBase/Firebird itself.

If you need to pack BLOB fields, write these two methods:

```
procedure PackBuffer(var Buffer: PChar; var BufSize: LongInt);  
var srcStream, dstStream: TStream;  
begin  
  srcStream := TMemoryStream.Create;  
  dstStream := TMemoryStream.Create;  
  try  
    srcStream.WriteBuffer(Buffer^, BufSize);  
    srcStream.Position := 0;  
    GZipStream(srcStream, dstStream, 6);  
    srcStream.Free;  
    srcStream := nil;  
    BufSize := dstStream.Size;  
    dstStream.Position := 0;  
    ReallocMem(Buffer, BufSize);  
    dstStream.ReadBuffer(Buffer^, BufSize);  
  finally  
    if Assigned(srcStream) then srcStream.Free;  
    dstStream.Free;  
  end;  
end;
```

```

procedure UnpackBuffer(var Buffer: PChar; var BufSize: LongInt);
var srcStream,dstStream: TStream;
begin
  srcStream := TMemoryStream.Create;
  dstStream := TMemoryStream.Create;
  try
    srcStream.WriteBuffer(Buffer^, BufSize);
    srcStream.Position := 0;
    GunZipStream(srcStream, dstStream);
    srcStream.Free;
    srcStream:=nil;
    BufSize := dstStream.Size;
    dstStream.Position := 0;
    ReallocMem(Buffer, BufSize);
    dstStream.ReadBuffer(Buffer^, BufSize);
  finally
    if assigned(srcStream) then srcStream.Free;
    dstStream.Free;
  end;
end;

```

Now we need to register the two methods before connecting to a database. Call the RegisterBlobFilter function. The first parameter value is BLOB field type (equals to -15), the second and third are packing and unpacking functions:

```
pFIBDatabase1.RegisterBlobFilter(-15, @PackBuffer, @UnpackBuffer);
```

You can also see the demo example BlobFilters for more details.

### ***Handling lost connections***

FIBPlus provides the developers with a unique feature of handling lost connections between the client and server. TpFIBDatabase and TpFIBErrorHandler components are responsible for handling this.

You can see the example ConnectionLost for demonstration. The notes below will explain how it works. The TpFIBDatabase has three special events

*AfterRestoreConnect* – fires if the connection was restored.

*OnLostConnect* – fires on the lost connection if any operation with the database caused an error. Here you can specify one of the three following actions (see the TOnLostConnectActions description) - close the application, ignore the error report or try to restore the connection.

*OnErrorRestoreConnect* – fires if the connection was not restored.

In the example when the connection is lost, the user has a number of choices. If the connection has been restored successfully, the corresponding message is shown. If the error occurs, you can count the number of restore attempts or do any other necessary actions.

We will provide you with more details about TpFIBErrorHandler in the corresponding event. If the connection is lost, the event handler suppresses the standard exception message.

```

procedure TForm1.dbAfterRestoreConnect(Database: TFIBDatabase);
begin
  MessageDlg('Connection restored', mtInformation, [mbOk], 0);
end;

procedure TForm1.dbErrorRestoreConnect(Database: TFIBDatabase;
  E: EFIBError; var Actions: TOnLostConnectActions);
begin
  Inc(AttemptRest);
  Label4.Caption:=IntToStr(AttemptRest);

```

```

    Label4.Refresh
end;

procedure TForm1.dbLostConnect(Database: TFIBDatabase; E: EFIBError;
var Actions: TOnLostConnectActions);
begin
    case cmbKindOnLost.ItemIndex of
        0: begin
            Actions := laCloseConnect;
            MessageDlg('Connection lost. TpFIBDatabase will be closed!',
                mtInformation, [mbOk], 0);
            end;
        1:begin
            Actions := laTerminateApp;
            MessageDlg('Connection lost. Application will be closed!',
                mtInformation, [mbOk], 0
            );
            end;
        2:Actions := laWaitRestore;
    end;
end;

procedure TForm1.pFibErrorHandler1FIBErrorEvent(Sender: TObject;
    ErrorValue: EFIBError; KindIBError: TKindIBError; var DoRaise: Boolean);
begin
    if KindIBError = keLostConnect then begin
        DoRaise := false;
        Abort;
    end;
end;

```

## *Other useful methods*

The TpFIBDatabase component has many useful methods. We will consider the most common of them.

### **How to execute simple SQL- queries**

If you need to execute a simple SQL query in order to get or set some application parameters, use the following methods:

```
function Execute(const SQL: string): boolean;
```

- executes an SQL-query, transferred in the SQL parameter, and returns True if the query was a success.

```
function QueryValue(const aSQL: string; FieldNo:integer; ParamValues:array of
variant; aTransaction:TFIBTransaction=nil):Variant; overload;
```

- it gets a field value with the FieldNo index as a result of executing aSQL in aTransaction. If you do not set the transaction, DefaultTransaction will be used. You can send parameters to the query. The value will be returned as a Variant variable. Use QueryValueAsStr to get a value as a string, and QueryValues – as an array. Remember that in this case the SQL must return not more than one string.

### **How to get generator values**

Use the following method to get generator values:

```
function Gen_Id(const GeneratorName: string; Step: Int64; aTransaction:
TFIBTransaction = nil): Int64;
```

### **How to get information about tables and fields**

```
procedure GetTableNames(TableNames: TStrings; WithSystem: Boolean);
procedure GetFieldNames(const TableName: string; FieldNames: TStrings;
```

```
WithComputedFields: Boolean = True);
```

The first method gets all table names and fills the TableNames list. The WithSystem parameter indicates whether to show system table names.

The second method takes TableName and fills FieldNames with the field names. The WithComputedFields parameter indicates whether to include COMPUTED BY fields.



## Working with transactions

A transaction is an operation of database transfer from one consistent state to another.

All operations with the dataset (data/metadata changes) are done in the context of a transaction. To understand special FIBPlus features completely you need to know about InterBase / FIBPlus transactions. Please read the topic «Working with Transaction» in [ApiGuide.pdf](#) for InterBase.

All the changes done in the transaction can be either committed (in case there are no errors) by Commit or rolled back (Rollback). Besides these basic methods TpFIBTransaction has their context saving analogues: CommitRetaining and RollbackRetaining, i.e. on the client side, these will not close a TpFibQuery or TpFibDataset.

To start the transaction you should call the StartTransaction method or set the Active property to True. To commit the transaction call Commit/CommitRetaining, to roll it back - Rollback/RollbackRetaining.

TpFIBQuery and TpFIBDataSet components have some properties which help to control transactions automatically. In particular they are: the TpFIBDataSet.AutoCommit property; the poStartTransaction parameter in TpFIBDataSet.Options; qoStartTransaction and qoCommitTransaction in TpFIBQuery.Options.

### *How to set transaction parameters*

Transaction parameters are not a trivial topic and require much explanation, so FIBPlus DevGuide won't cover the subject in detail. We highly recommend you to read [InterBase ApiGuide.pdf](#) to understand how transactions work.

Nevertheless in most cases you do not need to know about all peculiarities of transaction control at the API level. FIBPlus has a number of mechanisms which help developers' work easier. I.e. TpFIBTransaction has three basic transaction types: tpbDefault, tpbReadCommitted, tpbRepeatableRead. At design time you can also create special types of your own in the TpFIBTransaction editor and use them as internal ones. Set the transaction type to set its parameters::

tpbDefault – parameters must be set in TRParams

tpbReadCommitted – shows the ReadCommitted isolation level

tpbRepeatableRead – shows the RepeatableRead isolation level

### *Planning to use transactions in the application*

Efficient InterBase/Firebird applications depend heavily on correct transaction use. In a multi-generation architecture (record versioning) Update transactions retain record versions.

So in general try to make the Update transactions as short as possible. Read-only transactions can remain open because they do not retain versions.




Figure 2. Transaction Editor

### *How to use SavePoints*

InterBase/Firebird servers do not support nested transactions. But InterBase 7.X and Firebird 1.5 support SavePoints. FIBPlus realizes this functionality by three methods:

```

procedure SetSavePoint (const SavePointName:string) ;
procedure RollBackToSavePoint (const SavePointName:string) ;
procedure ReleaseSavePoint (const SavePointName:string) ;

```

The first method sets a save point with the SavePointName name. The second rolls the transaction back to SavePointName. The third releases SavePointName server resources.

### **SQL-query execution**

An application works with a database by issuing SQL instructions. They are used to get and modify data\metadata. FIBPlus has a special TpFIBQuery component responsible for SQL operator execution. This robust, light and powerful component can perform any actions with the database.

TpFIBQuery is very easy-to-use: just set the TpFIBDatabase component, fill in the SQL property and call any ExecQuery method (ExecQueryWP, ExecQueryWPS).

NOTE: The tpFIBQuery is not a TDataset descendant, so it does not act in exactly the same way or exhibit the same methods / properties as you would expect to find in a dataset. For the TDataset descendant, please refer to the TpFIBDataset.

The example below will show how to create TpFIBQuery dynamically at run-time and thus get data about clients.

```

var sql: TpFIBQuery;

sql := TpFIBQuery.Create (nil) ;
with sql do
try
  Database := db;
  Transaction := db.DefaultTransaction;
  SQL.Text := 'select first_name, last_name from customer';
  ExecQuery;
while not Eof do begin
  Memol.Lines.Add(
    FldByName['FIRST_NAME'].AsString+' '+
    FldByName['LASTST_NAME'].AsString);
  Next;

```

```

    end;
    sql.Close;
  finally
    sql.Free;
  end;

```

## ***How to transfer parameters***

Very often you need to use parameters in SQL queries. To this end FIBPlus has the Params property and TpFIBQuery methods ParamsCount, ParamByName. Besides there are some ExecWP methods (execute with parameters), which execute queries with preset parameters. It's really easy to use parameters as you can see from the code samples below:

```

sql.SQL.Text :=
  'select first_name, last_name from customer'+
  'where first_name starting with :first_name';

{ variant 1 }
sql.ParamByName('first_name').AsString := 'A';
sql.ExecQuery;

{ variant 2 }
sql.ExecWP('first_name', ['A']);

{ variant 3 }
sql.ExecWP(['A']);

```

## ***SQL- sections***

FIBPlus provides developers with a wide range of SQL query control capabilities. In particular they are SQL sections: a list of fields, conditions, grouping and sorting order, query execution plan. Effectively FIBPlus parses your SQL and identifies the following elements of an SQL statement. These simple string properties can be read and modified:

FieldsClause – has a field list;

MainWhereClause – has the main clause WHERE (see the details below);

OrderClause – has the clause «order by»;

GroupByClause – has the clause «group by»;

PlanClause – has the clause «plan».

FIBPlus also has unique features, such as macros and conditions – an extended mechanism of work with the WHERE clause. The sections below will provide you with more details about these features.

## ***Macros***

Macros help you to operate with the variable parts of your queries. This allows you to change and specify the queries without rewriting the query code.

The macro syntax is @@<MACROS\_NAME>[%<DEFAULT\_VALUE>|#]@

So macro is a specific order of symbols between the marks @@ and @. The parameter <MACROS\_NAME> is obligatory after @@. You can also set the default macro value after the symbol "%". Besides you can set the parameter # (not obligatory), it will make FIBPlus write parameter names in inverted commas.

Macros are used similar to parameters. This code example will demonstrate you this similarity:

```

Sql.SQL.Text := 'select * from @@table_clause@ where @@where_clause% 1=1@';
Sql.ExecWP(['CUSTOMER', 'FIRST_NAME STARTING WITH 'A'']);

```

Call the SetDefaultMacroValue method of the object parameter to set the default macro value.

Macro can also have a parameter. To search for it use the FindParam function, to set the parameter use the ParamByName method:

```
Sql.SQL.Text := 'select * from @@table_clause@ where @@where_clause% 1=1@';  
Sql.Params[0].AsString := 'CUSTOMER';  
Sql.Params[1].AsString := 'CUST_NO = :CUST_NO';  
if Assigned(Sql.FindParam('CUST_NO')) then  
    Sql.ParamByName('CUST_NO').AsInteger := 1001;  
Sql.ExecQuery;
```

You can also see the code example ServerFilterMacroses to get to know how to use macros for TpFIBDataSet

## Conditions

The mechanism of conditions is another option to change the variable part of your SQL-queries

You can set one or more parameter conditions for any SQL at design time or runtime. The built-in dialog shown in picture 3 is very convenient for these purposes.




Figure 3. Edit Conditions.

To make the condition active you should set the Active property to True.

```
pFIBQuery1.Conditions[0].Active := True;
```

or

```
pFIBQuery1.Conditions.ByName('by_customer').Active := True;
```

This is a sample code showing how to work with Conditions:

```
if pFIBQuery1.Open then pFIBQuery1.Close;  
pFIBQuery1.Conditions.CancelApply;
```

```

pFIBQuery1.Conditions.Clear;
if byCustomerFlag then
    pFIBQuery1.Conditions.AddCondition('by_customer', 'cust_no = 1001',
True);
pFIBQuery1.Conditons.Apply;
pFIBQuery1.Open;

```

TpFIBDataSet has two additional methods CancelConditions and ApplyConditions, which call correspondingly Conditions.Cancel and Conditions.Apply. This code sample for TpFIBDataSet is simpler than the previous.

```

with pFIBDataSet1 do begin
    if Active then Close;
    CancelConditioins;
    Conditions.Clear;
    if byCustomerFlag then Conditions.AddCondition('by_customer', 'cust_no =
1001', True);
    ApplyConditions;
    Open;
end;

```

You can also see the example ServerFilterConditions to get to know how to set conditions for TpFIBDataSet.

### ***Batch processing***

FIBPlus has internal methods for batch processing, called Batch methods. These methods can be helpful for replication between databases and import/export operations.

```

function BatchInput(InputObject: TFIBBatchInputStream) :boolean;
function BatchOutput(OutputObject: TFIBBatchOutputStream):boolean;
procedure BatchInputRawFile(const FileName:string);
procedure BatchOutputRawFile(const FileName:string;Version:integer=1);
procedure BatchToQuery(ToQuery:TFIBQuery; Mappings:TStrings);

```

The Version parameter is responsible for format compatibility with old file versions, created by FIBPlus BatchOutputXXX method. If Version = 1, FIBPlus uses an old principle of work with file versions: the external file keeps data ordered by SQL query fields. It is supposed that on reading data saved by the BatchInputRawFile method, parameters will have the same order in the reading SQL. The number of TpFIBQuery fields (the source of the data) must coincide to the number of TpFIBQuery parameters which will read the data. For string fields it is important to have the same length for the field being written and for the reading parameter whereas their names can differ.

If Version = 2, FIBPlus uses a new principle of writing data. Besides the data, the file also keeps system information about fields (name, type and length). On reading the data, it will be chosen by similar names. The order and number of fields in the writing TpFIBQuery can differ from those of parameters in the reading TpFIBQuery. Their types and length can also differ. Only names must coincide.

It's very easy to work with batch methods. We will show this using a simple example. The code below consists of three parts. The first saves data about clients into an external file, the second loads them into a database and the third shows how to change the data.

```

{ I }
pFIBQuery1.SQL := 'select EMP_NO, FIRST_NAME, LAST_NAME from CUSOMER';
pFIBQuery1.BatchOutputRawFile('employee_buffer.fibplus', 1);

{ II }
pFIBQuery1.SQL := 'insert into employees(EMP_NO, FIRST_NAME, LAST_NAME)'+
' values(:EMP_NO, :FIRST_NAME, :LAST_NAME)';

```

```

pFIBQuery1.BatchInputRawFile('employee_buffer.fibplus');

{ III }
pFIBQuery1.SQL := 'select EMP_NO, FIRST_NAME, LAST_NAME from CUSOMER';
pFIBQuery2.SQL := 'insert into tmp_employees(EMP_NO, FIRST_NAME, LAST_NAME)'+
  ' values(:EMP_NO, :FIRST_NAME, :LAST_NAME)';

mapStrings.Add('EMP_NO=EMP_NO');
mapStrings.Add('FIRST_NAME=FIRST_NAME');
mapStrings.Add('LAST_NAME=LAST_NAME');

pFIBQuery1.BatchToQuery(pFIBQuery2, mapStrings);

```

We will discuss more about batch processing when talking about TpFIBDataSet as it also has some batch processing methods.

The OnBatchError event occurs in case of incorrect processing. Using the parameter BatchErrorAction (TBatchErrorAction = (beFail, beAbort, beRetry, beIgnore)) in the code of this event you can decide what to do in this case.

### ***Execution of stored procedures***

Execution of stored procedures is very similar to query execution. You only need to write 'execute procedure some\_proc(:proc\_param)' in the SQL text or 'select \* from some\_proc(:proc\_param)' for selectable procedures (i.e. those that return a result set).

If non-selectable procedure returns any results you can get them after the query executing using the Fields property.

```

Sql.SQL.Text := 'execute procedure some_proc(:proc_param)';
Sql.ExecWP([25]);
ResultField1 := Sql.Fields[0].AsInteger;

```

This feature makes FIBPlus different from BDE, ADO and other libraries where input data are also available through parameters.

Besides, FIBPlus enables developers to execute stored procedures by using the TpFIBStoredProc component. TpFIBStoredProc is a direct TpFIBQuery descendant with the StoredProcName property. **It is recommended to use TpFIBStoredProc to execute non-selectable procedures.**

### ***How to execute DDL(Data Definition Language) commands.***

DDL is the subset of SQL that allows you to change the structure of the database, e.g. to create tables.

Besides SQL-operators TpFIBQuery helps to execute DDL-commands. In order to be able to execute a DDL-command you need to set the ParamsCheck property to False. New FIBPlus versions support macros for DDL.

### ***Recurrent use of queries***

All client libraries including FIBPlus have to transfer the complete query text in order to prepare a query for execution. Once you have a prepared query it is enough to transfer only handle and parameter values. FIBPlus knows when you have changed the SQL in a TpFIBQuery, so it will only prepare when it is required. By TpFIBPlus handling the preparing of queries for you, it ensures re-using the same component for the same query, merely changing the parameters, will offer optimum performance.

If however, your application does not lend itself to using the same query component for the same query, FIBPlus offers a query pool mechanism. If there are numerous similar query requirements in your application you can use methods from pFIBCacheQueries.pas to manage the recurrent use:

```
function GetQueryForUse (aTransaction: TFIBTransaction; const SQLText:
string): TpFIBQuery;
procedure FreeQueryForUse (aFIBQuery: TpFIBQuery);
```

You don't need to create TpFIBQuery instances. Being called for the first time the GetQueryForUse procedure will create a TpFIBQuery instance and then will return a link to the existing component when you execute the same query again and again. As you see, on every recurrent procedure call FIBPlus will use the prepared query and thus transfer the query text to the server only once. When you don't need the recurrent query anymore (when the query results are obtained from the TpFIBQuery component) you should call the FreeQueryForUse method. Such FIBPlus mechanism is used for internal purposes i.e. on calling generators to get primary key values. You can use these methods in your applications to optimize network traffic.

## Work with datasets

The TpFIBDataSet component is responsible for work with datasets. It is based on the TpFIBQuery component and helps to cache selection results. TpFIBDataSet is a TDataSet descendant so it supports all TDataSet properties, events and methods. To get more information about TDataSet please read Delphi/C++Builder help manuals.

### *Basic principles of work with datasets*

TpFIBDataSet enables developers to select, insert, update and delete data. All these operations are executed by TpFIBQuery components in TpFIBDataSet.

To select data you set the SelectSQL property. It's similar to setting the SQL property of the QSelect component (TpFIBQuery type). Define the InsertSQL.Text property to insert data, UpdateSQL.Text to update, DeleteSQL.Text to delete and RefreshSQL.Text to refresh the data.

We will use a demo database employee.gdb (or .fdb for Firebird) to show how to write Select SQL and get a list of all employees. We will write all queries in InsertSQL, UpdateSQL, etc.

```
with pFIBDataSet1 do begin
  if Active then Close;
  SelectSQL.Text :=
    'select CUST_NO, CUSTOMER, CONTACT_FIRST, CONTACT_LAST from CUSTOMER';

  InsertSQL.Text :=
    'insert into CUSTOMER(CUST_NO, CUSTOMER, CONTACT_FIRST, CONTACT_LAST )'+
    ' values (:CUST_NO, :CUSTOMER, :CONTACT_FIRST, :CONTACT_LAST)';

  UpdateSQL.Text :=
    'update CUSTOMER set CUSTOMER = :CUSTOMER, '+
    'CONTACT_FIRST = :CONTACT_FIRST, CONTACT_LAST = :CONTACT_LAST '+
    'where CUST_NO = :CUST_NO';

  DeleteSQL.Text := 'delete from CUSTOMER where CUST_NO = :CUST_NO';

  RefreshSQL.Text :=
    'select CUST_NO, CUSTOMER, CONTACT_FIRST, CONTACT_LAST ' +
    'from CUSTOMER where CUST_NO = :CUST_NO';
```

```
Open;  
end;
```

To open TpFIBDataSet either execute Open/OpenWP methods or set the Active property to True. To close TpFIBDataSet call the Close method.

Don't be concerned by seeing a lot of code lines, all these queries can be automatically created by the TpFIBDataSet editor. You can call it from the component context menu, as shown in picture 4.




Figure 4. TpFIBDataSet SQL Editor

The example DataSetBasic demonstrates basic use of editable TpFIBDataSet.

### ***Automatic generation of Update queries***

Besides TpFIBDataSet SQL editor, FIBPlus can generate all Update queries at run-time in a more effective way than at design-time.

For this purpose use the AutoUpdateOptions property. This group of setting is very important as it makes the coding process simpler and more convenient.

```
TAutoUpdateOptions= class (TPersistent)  
  property AutoParamsToFields: Boolean .. default False;  
  property AutoRewriteSqls: Boolean .. default False;  
  property CanChangeSQLs: Boolean .. default False;  
  property GeneratorName: string;  
  property GeneratorStep: Integer .. default 1;  
  property KeyFields: string;  
  property ParamsToFieldsLinks: TStrings;  
  property SeparateBlobUpdate: Boolean .. default False;  
  property UpdateOnlyModifiedFields: Boolean .. default False;  
  property UpdateTableName: string;  
  property WhenGetGenID: TWhenGetGenID .. default wgNever;  
end;
```



```
TWhenGetGenID= (wgNever, wgOnNewRecord, wgBeforePost) ;
```

*AutoRewriteSQLs* - If there are empty *SQLText* properties for *InsertSQL*, *UpdateSQL*, *DeleteSQL*, *RefreshSQL* they will be automatically generated by the *SelectSQL*, *KeyFields* and *UpdateTableName* properties.

*CanChangeSQLs* informs that non-empty queries can be rewritten.

*GeneratorName* sets the generator name and *GeneratorStep* sets the generator step.

*KeyFields* contains a list of key fields.

*SeparateBlobUpdate* manages BLOB-field writing in a database. If *SeparateBlobUpdate* is set to True at first a record will be saved without a BLOB-field and then if the operation is a success the BLOB-fields will be also written to the database.

If *UpdateOnlyModifiedFields* is set to True and if *CanChangeSQLs* is set to True, a new SQL query will be automatically created for each modifying operation. This SQL query will contain only fields, which were changed.

*UpdateTableName* must contain a name of the modified table.

*WhenGetGenId* enables developers to set a mode of using a generator to form a primary key. Either not to generate the primary key, to generate it on adding a new record, or before Post.

So due to *AutoUpdateOptions* settings FIBPlus helps not to generate modifying queries at design-time and for this at run-time. To use this feature you just need to write a name of modified table and the key field.

The code below is taken from the example *AutoUpdateOptions*. You can use this feature both at design-time and run-time:

```
pFIBDataSet1.SelectSQL.Text := 'SELECT * FROM EMPLOYEE';  
pFIBDataSet1.AutoUpdateOptions.AutoRewriteSqls := True;  
pFIBDataSet1.AutoUpdateOptions.CanChangeSQLs := True;  
pFIBDataSet1.AutoUpdateOptions.UpdateOnlyModifiedFields := True;  
pFIBDataSet1.AutoUpdateOptions.UpdateTableName := 'EMPLOYEE';  
pFIBDataSet1.AutoUpdateOptions.KeyFields := 'EMP_NO';  
pFIBDataSet1.AutoUpdateOptions.GeneratorName := 'EMP_NO_GEN';  
pFIBDataSet1.AutoUpdateOptions.WhenGetGenID := wgBeforePost;  
pFIBDataSet1.Open;
```

## ***Local sorting***

FIBPlus has local sorting methods, which helps you to order TpFIBDataSet buffer in any possible way as well as the ability to remember and restore sorting after TpFIBDataSet has been closed and reopened. In addition FIBPlus enables developers to use the mode where all newly inserted and changed records will be placed to the correct buffer position according to the sorting order. To sort TpFIBDataSet buffer you should call any of three following methods.

```
procedure DoSort (Fields: array of const; Ordering: array of Boolean);  
virtual;  
procedure DoSortEx (Fields: array of integer; Ordering: array of Boolean);  
overload;  
procedure DoSortEx (Fields: TStrings; Ordering: array of Boolean); overload;
```

The first parameters defines the list of fields, the second is an array of sorting orders. If you set the sorting order to True it will be ascending, to False – descending.

Here are two examples of the same sorting by two fields in the ascending order:

```
pFIBDataSet1.DoSort(['FIRST_NAME', 'LAST_NAME'], [True, True]);  
or  
pFIBDataSet1.DoSortEx([1, 2], [True, True]);
```

You can get the current sorting order by checking the following properties:

```
function SortFieldsCount: integer;  
returns the sorting field number  
function SortFieldInfo (OrderIndex: integer): TSortFieldInfo -  
returns information about sorting in the OrderIndex position.  
function SortedFields: string  
returns a string with sorting fields enumerated by '  
TSortFieldInfo = record  
  FieldName: string; //field name  
  InDataSetIndex: Integer; //whether included into index  
  InOrderIndex: Integer; //included ...  
  Asc: Boolean; //True, if ascendant  
  NullsFirst: Boolean; //True, if Null values are the first  
end;
```

Set poKeepSorting property to True to put records to the right buffer position on inserting and editing. Use the psGetOrderInfo option if you want to TpFIBDataSet automatically use the local sorting order defined in the ORDER BY query statement.

Set psPersistentSorting option to True in order to keep the sorting order on TpFIBDataSet reopening. Be careful, if you have huge selections this feature will not be effective because at first the buffer will retrieve all records from the server and only then sort them.

## **Sorting of national symbols**

Two parameters are responsible for correct sorting of national symbols: CHARSET (set of symbols) and COLLATION (sorting order). Even if you correctly set these parameters in a database and queries TpFIBDataSet may sort them incorrectly. The point is that default local sorting uses the simplest method without comparing national symbols. As if NONE charset is set.

TpFIBDataSet can sort national symbols according to the national charset. If you use the OnCompareFieldValues event you can compare char fields in an alternative way. There are three standard methods for Ansi-sorting:

```
protected function CompareFieldValues(Field:TField;const
S1,S2:variant):integer; virtual;
```

```
public function AnsiCompareString(Field:TField;const val1, val2: variant):
Integer;
```

```
public function StdAnsiCompareString(Field:TField;const S1, S2: variant):
Integer;
```

AnsiCompareString is case-sensitive, and StdAnsiCompareString is not.

```
pFIBDataSet1.OnCompareFieldValues := pFIBDataSet1.AnsiCompareString;
pFIBDataSet1.OnCompareFieldValues := pFIBDataSet1.StdAnsiCompareString;
```

You can set any standard methods for CompareFieldValues as default or write your own method if necessary.

### ***Local filtering***

In contrast to IBX, TpFIBDataSet has full support of local filtering. It supports the Filter and Filtered properties and the OnFilterRecord event.

The table below shows the additional operators which can be used in the Filter property.

<b><i>Operation</i></b>	<b><i>Description</i></b>
<	Less than.
>	Greater than.
>=	Greater than or equal.
<=	Less than or equal.
=	Equal to
<>	Not equal to
AND	Logical AND
NOT	Logical NOT
OR	Logical OR
IS NULL	Tests that a field value is null
IS NOT NULL	Tests that a field value is not null
+	Adds numbers, concatenates strings, adds number to date/time values
-	Subtracts numbers, subtracts dates, or subtracts a number from a date
*	Multiplies two numbers
/	Divides two numbers
Upper	Upper-cases a string
Lower	Lower-cases a string
Substring	Returns the substring starting at a specified position
Trim	Trims spaces or a specified character from front and back of a string
TrimLeft	Trims spaces or a specified character from front of a string
TrimRight	Trims spaces or a specified character from back of a string
Year	Returns the year from a date/time value
Month	Returns the month from a date/time value
Day	Returns the day from a date/time value
Hour	Returns the hour from a time value

<i><b>Operation</b></i>	<i><b>Description</b></i>
Minute	Returns the minute from a time value
Second	Returns the seconds from a time value
GetDate	Returns the current date
Date	Returns the date part of a date/time value
Time	Returns the time part of a date/time value
Like	Provides pattern matching in string comparisons
In	Tests for set inclusion
*	Wildcard for partial comparisons.

If the Filter property is set to True and is not active, you can use the following methods to navigate on records which satisfy the filtering conditions:

*FindFirst*

*FindLast*

*FindNext*

*FindPrior*

In case of huge data sets we recommend you to use the server filtering. It can be realized by macros and conditions mechanisms.

To know more about local filtering see the example LocalFiltering.

**Important:** use the VisibleRecordCount function instead of RecordCount to get the number of records by the Filter condition.

## ***Data search***

TpFIBDataSet supports Locate, LocateNext and LocatePrior methods, which are described in a standard Delphi/C++Builder help manual.

Besides FIBPlus has some specific analogues which have some important advantages<sup>^</sup>

```
function ExtLocate(const KeyFields: String; const KeyValues: Variant;
Options: TExtLocateOptions): Boolean;
```

```
function ExtLocateNext(const KeyFields: String; const KeyValues: Variant;
Options: TExtLocateOptions): Boolean;
```

```
function ExtLocatePrior(const KeyFields: String; const KeyValues: Variant;
Options: TExtLocateOptions): Boolean;
```

TExtLocateOptions = (eloCaseInsensitive, eloPartialKey, eloWildCards, eloInSortedDS, eloNearest, eloInFetchedRecords)

eloCaseInsensitive to ignore the case;

eloPartialKey partial coincidence

eloWildCards search by wild cards (similar to LIKE operator);

eloInSortedDS search in a sorted dataset (influences the search speed);

eloNearest only together with eloInSortedDS. It is placed where “must be”;

eloInFetchedRecords search only in fetched records.

## ***Master and detail datasets***

Besides a standard TDataSet Master-detail mechanism FIBPlus has an additional group of options DetailCondition, described as:

```
TDetailCondition=(dcForceOpen,dcIgnoreMasterClose,dcForceMasterRefresh,  
dcWaitEndMasterScroll);
```

```
TDetailConditions= set of TDetailCondition;
```

*dcForceOpen* if it is active, the detail TpFIBDatSets will be open on opening the master ;

*dcIgnoreMasterClose* if it is active, the detail TpFIBDataSet won't close on closing the master;

*dcForceMasterRefresh* if it is active, the current master record will be refreshed on refreshing the the detail TpFIBDataSet;

*dcWaitEndMasterScroll* if it is active, on scrolling at master TpFIBDataSet will wait a little before reopening the detail. This option helps to avoid useless operations if the master navigation is simple.

### ***Pessimistic locking***

Standard record changing behavior of InterBase/Firebird servers is optimistic locking. If two or more users edit the same record at the same time, only the first modification is written to the database, and the second receives an exception error.

As a rule if you need a pessimistic locking in InterBase/Firebird, you use a «dummy update». It means that the record is updated i.e. by the primary key before record editing:

```
update customer set cust_no = cust_no where cust_no = :cust_no
```

Then the actual record is automatically fetched from the server. This behaviour helps to guarantee that the record won't be updated from another transaction before the end of the dummy update transaction.

FIBPlus manages this process automatically. You need to activate the psProtectedEdit option or use the the TpFIBDataSet.LockRecord method.

The demo example ProtectedEditing demonstrates how this feature works.

### ***Work in the confined local buffer mode – for huge datasets and random access***

The mode was first suggested by Sergey Spirin in gb\_Datasets components. Now FIBPlus is also capable of this feature (since version 6.0). It enables navigation of TpFIBDataSet without fetching all the records returned by the query. In fact it is simulation of random access to records by means of supplementary queries. The technology sets a number of query requirements. In particular, one of the obligatory requirements is use of ORDER BY in SelectSQL. First, in ORDER BY it's important to make the combination of field values unique. Second, to speed up data transfer time it's better to have two indices - ascending and descending - for this field combination.

This simple example illustrates the technology. Having such a query in SelectSQL:

```
SELECT * FROM TABLE  
ORDER BY FIELD
```

You may get some first records, fetching them successively. To see the last records immediately, you may execute an additional query with descending sorting instead of querying all the records from the server,:

```
SELECT * FROM TABLE  
ORDER BY FIELD DESC
```

Obviously successive fetching of several records will result the last records (in relation to the initial query). Similar queries are for exact positioning on any record, as well as on records below and above the current one:

```
SELECT * FROM TABLE  
WHERE (FIELD = x)
```

```
SELECT * FROM TABLE  
WHERE (FIELD < x)  
ORDER BY FIELD DESC
```

```
SELECT * FROM TABLE  
WHERE (FIELD > x)  
ORDER BY FIELD
```

To carry out this technology TpFIBDataSet has a new property:

property CacheModelOptions:TCacheModelOptions, where

```
TCacheModelOptions = class(TPersistent)  
property BufferChunks: Integer ;  
property CacheModelKind: TCacheModelKind ;  
property PlanForDescSQLs: string ;  
end;
```

BufferChunks replaces the existing property BufferChunks of TpFIBDataSet. The TCacheModelKind type can have a cmkStandard value for the standard local buffer work and a cmkLimitedBufferSize value for the new technology of the confined local buffer. The buffer size is a number of records set in BufferChunks.

The PlanForDescSQLs property enables to set a separate plan for queries with descending sorting.

**Note:** when using the technology of the confined local buffer,

- You must not activate the CachedUpdate mode;
- The RecNo property will return incorrect values;
- Local filtering will not be supported;
- Work with BLOB-fields may be not stable in the present version;
- You should activate the psGetOrderInfo option in PrepareOptions.

### ***Work with the internal dataset cache***

TpFIBDataSet has several special methods for work with its internal record cache (this allows you to excute prior without having to go back to the server). Actually these methods make TpFIBDataSet an analogue of TClientDataSet oriented at InterBase. Its only difference from TClientDataSet is that there must be a connection with the database and SelectSQL must have a correct query. Despite these restrictions the mechanism is very flexible and helps to realize numerous “non standard” things. For example this query will select one Integer field and one String:

```
select cast(0 as integer) some_id, cast(" as varchar(255)) some_name  
from RDB$DATABASE.
```

This would, if executed, return one row with an integer field (set to 0) and a string field with an empty string.

You can open this TpFIBDataSet by calling the CacheOpen method. Then you can use the following methods:

```
procedure CacheModify(aFields: array of integer; Values: array of Variant;
KindModify: byte );
procedure CacheEdit(aFields: array of integer; Values: array of Variant);
procedure CacheAppend(aFields: array of integer; Values: array of Variant);
overload;
procedure CacheAppend(Value: Variant; DoRefresh: boolean = False); overload;
procedure CacheInsert(aFields: array of integer; Values: array of Variant);
overload;
procedure CacheInsert(Value: Variant; DoRefresh: boolean = False); overload;
procedure CacheRefresh(FromDataSet: TDataSet; Kind: TCachRefreshKind ;
FieldMap: Tstrings);
procedure CacheRefreshByArrMap( FromDataSet: TDataSet; Kind:
TCachRefreshKind; const SourceFields, DestFields: array of string )
```

to add a record execute:

```
pFIBDataSet1.CacheInsert([0,1],[255, 'string1'])
```

to modify:

```
pFIBDataSet1.CacheModify([0,1],[255, 'string1'])
```

to delete from cache, call CacheDelete;

The CacheRefresh and CacheRefreshByArrMap methods enable to refresh a record on basis of data from another TpFIBDataSet.

All these operations do not change the database as they are executed in TpFIBDataSet cache.

Sometimes you can also use this technique in a standard mode. For example when you need to insert a record using some complex stored procedure, which returns the code of the inserted record, and then to show the code in TpFIBDataSet. You can insert the code and call the Refresh method:

```
id := SomeInsertByProc;
pFIBDataSet1.CacheInsert([0], [1]);
pFIBDataSet1.Refresh;
```

In addition you could also delete some non-existing records from cache without having to refresh the query.